```
∇Z← F N
 ⍎(N=0)/'→0, 0ρZ←1'
 Z←N× F N-1
∇

      F 4
24
```

In direct definition:

```
      PLUS: α + ω

      3 PLUS 4
7
      F: ω × F ω-1 : ω=0 : 1
F
      F 4
24
```

The left and right arguments are denoted α and ω. The recursive definition of the factorial should be read: "The factorial of ω is ω times the factorial of ω-1 unless ω equals zero, in which case the factorial of ω is 1."

To illustrate the advantage of Iverson's method, consider the problem of cluster analysis. Each entity, described by *n* variables, can be considered a point in n-dimensional space, and we are required to compute the distance between each point and all the others. If n is 2, the data are given in a matrix of two columns. We then represent each entity as a point, with coordinates *x* and *y*, plotting the points on a scatter diagram. The theorem of Pythagoras lets us determine the distance between any two points, and the results complete a square matrix. This *similarity matrix* gives the *closeness* of each entity to every other one based on all measured properties. The matrix is symmetric with zeros on the diagonal. In APL the algorithm automatically extends to higher dimensions.

Hellerman used this as an example of APL notation, in a book that (in both of its editions) is a landmark in the history of APL.[28] His solution is as follows:

```
∇Z←DISTANCE P;N;I;J
 N←(ρP)[0]
 D←(N,N)ρ0
 J←0
LO:I←0
L1:D[I;J]←(P[I;]-P[J;])
  +.×(P[I;]-P[J;])
 →(N>I←I+1)/L1
 →(N>J←J+1)/LO
∇
```

Direct definition allows this to be expressed in a single line:

```
DSQ: (0 2 1 2⍉ω∘.-ω)+.*2
```

If this line seems strange or unduly terse to someone new to APL, I would point out that if we already know how to add, subtract, and square numbers, there are only three APL functions to learn: inner and outer products and dyadic transposition. I remember Adin Falkoff saying that good notation cannot make an inherently recondite concept easy, but it can remove unnecessary impediments by expressing the concept in as simple a manner as possible: Einstein's $E = m \times c^2$ is a simple statement of a relationship that probably can be fully understood by very few.

For a further illustration consider eight statistical functions, first in standard APL notation:

```
∇Z←MEAN X
 Z←(+/X) ÷ 0⊥ρX
∇

∇Z←DEV X
 Z←X- (MEAN X)∘.+ (0⊥ρX)ρ0
∇

∇Z←SS X
 Z←(DEV X)+.*2
∇

∇Z←VAR X
 Z←(SS X)÷ ¯1+0⊥ρX
∇

∇Z←SD X
 Z←(VAR X)*0.5
∇

∇Z←SP X;M
 Z←M+.× ⍉M← DEV X
∇

∇Z←COV X
 Z←(SP X)÷ ¯1+0⊥ρX
∇

∇Z←COR X;S
 Z←(COV X)÷S∘.×S←SD X
∇
```

They define the means, deviations from the means, sums of squares of the deviations, variances, standard deviations, sums of cross products, covariances, and correlation coefficients.